




# GUIDE DE PROGRAMMATION

Scans de Marché (ProScreener)



# SOMMAIRE

 <b>Présentation de ProScreener</b>	<b>1</b>
 <b>Chapitre I : Les notions fondamentales</b>	<b>2</b>
➔ Accéder à ProScreener.....	2
➔ Utiliser la fenêtre Palmarès.....	6
 <b>Chapitre II : La programmation sur ProScreener</b>	<b>7</b>
➔ Instruction de recherche et de filtres.....	7
> Utiliser plusieurs critères de tri.....	9
> Formater les données de sortie.....	9
➔ Anticipation du volume.....	10
➔ Recherche multi-périodes.....	10
➔ Recherche multi-valeurs.....	11
 <b>Chapitre III : Applications pratiques</b>	<b>13</b>
➔ Exemples simples.....	13
> RSI 1heure : survente.....	13
> RSI 1heure : surachat.....	13
> Croisement de moyennes mobiles haussier.....	13
> Croisement de moyennes mobiles baissier.....	14
➔ Exemples plus élaborés.....	14
> RSI et retournement haussier.....	14
> RSI et retournement baissier.....	15
> Avalement haussier.....	15
> Avalement baissier.....	16
> Triple écran haussier.....	16
> Triple écran baissier.....	17
 <b>Glossaire</b>	<b>18</b>

*Avertissement : ProRealTime n'exerce pas le service de Conseil en Investissement Financier. Ce document n'est en aucun cas une offre de conseil en investissement ni une incitation quelconque à acheter ou vendre des instruments financiers. Les exemples présentés dans ce manuel sont à but pédagogique. Pour votre propre trading, vous êtes entièrement libre dans le choix de vos critères. Les performances passées ne présagent pas de l'avenir. Tout système de trading peut vous exposer à un risque de perte supérieur à votre investissement initial.*

## Présentation de ProScreener

ProScreener est un outil de scan très performant qui vous permettra de passer en revue des marchés entiers :

- à la recherche des valeurs respectant une ou plusieurs conditions que vous aurez définies.
- dans une ou plusieurs unités de temps (ex : 1 minute et 1 heure)
- en temps-réel ou à la clôture de chaque barre avec une précision tick par tick

ProScreener intègre le langage de programmation ProBuilder (dont il vous est conseillé de lire au préalable [le manuel](#)) avec des extensions s'appliquant exclusivement aux conditions de filtrage. Les recherches en temps-réel ou à clôture de la barre, peuvent s'appliquer aux unités de temps suivantes :

- 1 minute
- 2 minutes
- 3 minutes
- 5 minutes
- 10 minutes
- 15 minutes
- 30 minutes
- 1 heure
- 2 heures
- 3 heures
- 4 heures
- Journalier
- Hebdomadaire
- Mensuel
- Trimestriel
- Annuel

ProScreener utilise les 256 derniers chandeliers (1024 pour les plateformes premium) de l'unité de temps choisie afin de réaliser les calculs de votre code.

Les résultats d'une recherche ProScreener sont affichés par une liste des 50 ou 100<sup>1</sup> meilleures valeurs (choisies parmi les résultats) sur la base du critère de tri défini.

Ce document s'inscrit dans le prolongement du [manuel ProBuilder](#) mais peut être lu indépendamment. Grâce à une approche théorique guidée et des exemples concrets, vous acquerez au fil de votre lecture les bases de la programmation d'indicateurs qui vous permettrons, à partir des commandes spécifiques à ProScreener, de définir des filtres sur les différents marchés disponibles sur la plateforme. Vous trouverez, à la fin de ce dernier, un Glossaire qui vous donnera une vue de l'ensemble des commandes utilisables dans ce module.

Pour toutes questions complémentaires sur le fonctionnement du ProScreener vous pouvez les poser auprès de notre communauté ProRealTime sur le [forum ProRealCode](#), vous y retrouverez aussi une [documentation en ligne](#) avec de nombreux exemples.

En vous souhaitant nos meilleurs vœux de réussite, bonne lecture !


---

<sup>1</sup> En fonction de la version ProRealTime utilisée

## Chapitre I : Les notions fondamentales

### Accéder à ProScreener

Vous pouvez accéder au module ProScreener en cliquant sur "Affichage" puis "ProScreener", comme montré ci-dessous :




Apparaîtra alors la fenêtre ProScreener, où il est possible de sélectionner un code déjà créé ou bien de procéder à la création d'un nouveau code.

Pour ce faire, cliquez sur la clé pour ouvrir la "Gestion ProScreeners", puis cliquer sur « Créer » ou « Modifier » afin d'accéder à la fenêtre de programmation, qui propose le choix entre :

- la "Programmation assistée", qui permet de définir les conditions du critère de recherche sans programmer (pour en savoir plus sur cette fonctionnalité, nous vous conseillons la vision de la vidéo, "[Créer un ProScreener sans écrire une ligne de code](#)").
- la "Création par programmation", qui permet de créer des codes plus complexes et de définir les paramètres d'application.

La fenêtre de Création par programmation se compose de 4 sections :

1. Zone de programmation
2. Type de sélection
3. Période considérée et sens du tri
4. Choix des listes/groupes




Dans la **première** section vous pouvez :

- Programmer directement un **ProScreener** dans la zone de texte
- Utiliser la fonction d'aide "Insérer Fonction", qui permet de retrouver dans une nouvelle fenêtre la bibliothèque des fonctions disponibles, séparées en neuf catégories, afin de vous aider contextuellement lors de la programmation.

Prenons comme exemple l'instruction "**SCREENER**", qui permet d'effectuer la recherche des conditions (disponible dans la section "Commandes ProScreener").

Sélectionnez donc le mot "**SCREENER**" et cliquez sur "Ajouter" : la commande s'ajoutera à la zone de programmation.



La commande "**SCREENER**" vous permet de définir quelle(s) condition(s) rechercher.

Supposons que l'on souhaite rechercher toutes les valeurs du SRD France dont l'ouverture de la barre (période) en cours soit supérieure à la clôture de la barre précédente . Nous écrivons donc :

```
c1 = (Open > Close[1])
SCREENER[c1]
```

Une fois le code défini, nous allons choisir dans la **deuxième** section, le type de sélection sur laquelle sera appliquée le screener : un groupe, une liste personnalisée ou un secteur.

La **troisième** section permet de définir la période utilisée dans la recherche du ProScreener. En effet, chaque cours pouvant être visualisé selon une période différente, une figure repérée sous une certaine vue pourrait ne pas exister dans une autre. On pourra aussi y définir le tri des résultats. Avec des conditions peu restrictives, les résultats d'un ProScreener peuvent être nombreux (plusieurs milliers par exemple sur le NASDAQ). Dans de tels cas, il convient de filtrer les résultats selon un ou plusieurs critères aux choix et d'afficher :

- les 50 (ou 100<sup>2</sup>) valeurs les plus élevées (du critère de tri)
- les 50 (ou 100<sup>2</sup>) valeurs les plus faibles (du critère de tri)

La **quatrième** section concerne le choix du/des groupes, listes ou secteurs utilisé(s) pour la recherche.

Par exemple : imaginons que votre programme recherche les volumes supérieurs à 20000 sur les valeurs du SRD en journalier. Le nombre de résultats suite à la recherche devrait logiquement excéder 50 résultats. Paramétrer le critère de tri par exemple sur le % de variation des titres vous permettra d'afficher parmi les résultats les 50 valeurs ayant les variations les plus élevées ou les plus faibles.

Une fois ces 4 paramètres définis, cliquez sur "Exécuter ProScreener" pour lancer la recherche. Si des valeurs entrent dans vos conditions, les résultats sont affichés dans une liste comme ci-dessous

Nom	%Var	Dernier
HERMES INTL	+1,43 %	1 494,0
LVMH	+1,82 %	706,7
KERING	+1,82 %	492,40
LOREAL	+0,98 %	340,40
TELEPERFORMANCE	+2,42 %	228,5
PERNOD RICARD	+0,95 %	186,65
ESSILORLUXOTTICA	+0,67 %	172,10
CAPGEMINI	+1,48 %	160,95
AIR LIQUIDE	+0,53 %	135,72
SCHNEIDER ELECTRIC	+0,84 %	133,86
SAFRAN	+0,94 %	119,66
THALES	-1,33 %	118,40
AIRBUS	+1,18 %	115,20
VINCI	+0,46 %	95,39
SANOFI	+0,49 %	92,13
LEGRAND	+0,98 %	75,96
EUROFINS SCIENT.	+0,88 %	66,60
PUBLICIS GROUPE SA	+1,31 %	60,44
TOTALENERGIES	+0,53 %	60,34
BNP PARIBAS ACT.A	+2,18 %	56,13
UNIBAIL-RODAMCO-WE	+0,73 %	51,10
DANONE	+1,09 %	50,12

✓ Nb résultats : 40 (max 100 affichés) Quantité d'historique : 1024

<sup>2</sup> En fonction de votre version de ProRealTime

## Utiliser la fenêtre Palmarès

En plus du ProScreener, ProRealTime met à votre disposition la fenêtre palmarès. Les palmarès sont des fenêtres de scan en temps réel, dont les critères sont prédéfinis (contrairement au ProScreener, qui est totalement personnalisable). Elles vous permettent des recherches simultanées (exemple : variation à la hausse dans une fenêtre, variations à la baisse dans l'autre fenêtre).

Les critères de recherche proposés dans le Palmarès appartiennent à 4 catégories et permettent de détecter :

- Les variations et gaps
- Les plus amples variations de pré-ouverture
- Les principales figures de chandeliers
- Les valeurs de Spread et de volume du carnet d'ordre

Palmarès Variations			
France SRD		Variation Hausse	Hier
Nom	%Var Veille	Dernier	
GENSIGHT BIOLOGICS	+7,91 %	3,710	
GENERAL ELECTRIC	+5,59 %	79,4	
TRANSGENE	+4,66 %	1,798	
VALNEVA	+3,85 %	6,854	
SCHLUMBERGER	+3,67 %	50,8	
RALLYE	+3,43 %	2,865	
CEGEDIM	+3,33 %	15,50	
SWORD GROUP	+3,03 %	40,80	
JACQUET METALS	+2,70 %	17,48	
AIR FRANCE -KLM	+2,67 %	1,2860	
BALYO	+2,66 %	0,4250	
STMICROELECTRONICS	+2,60 %	34,285	
ERYTECH PHARMA	+2,58 %	0,3900	
TELEPERFORMANCE	+2,38 %	228,4	
MANITOU BF	+2,33 %	26,35	
LECTRA	+2,32 %	35,25	
FAURECIA	+2,32 %	15,685	
CREDIT AGRICOLE	+2,28 %	10,214	
SCOR SE	+2,26 %	22,16	
ENGIE	+2,21 %	13,856	
DASSAULT SYSTEMES	+2,19 %	34,515	
BNP PARIBAS ACT.A	+2,18 %	56,13	
KAUFMAN ET BROAD	+2,18 %	28,10	

✓ Nb résultats : 100

Afin d'en savoir plus sur l'utilisation du Palmarès, nous vous conseillons la vision de la vidéo "[Comment utiliser les palmarès](#)"

## Chapitre II : La programmation sur ProScreener

Dans ce chapitre vous seront illustrées les 5 commandes ProBuilder propres au ProScreener. Vous pouvez retrouver ces commandes également dans la Bibliothèque des fonctions, à la section "Commandes ProScreener". Nous passerons en revue :

- Instruction de recherche et de filtre
- Anticipation du volume
- Recherche multi-période
- Recherche multi-valeur

### Instruction de recherche et de filtres

---

La commande "**SCREENER**" permet de lancer la recherche. Elle est l'équivalent de "**RETURN**" pour ProBuilder et, comme cette dernière, peut être suivie d'une panoplie de fonctions que nous allons étudier ensemble.

"**SCREENER**" s'utilise de la façon suivante :

```
SCREENER[Condition]
```

Prenons un exemple :

```
c1 = (Close < BollingerDown[10](Close))
SCREENER[c1]
```

Cherchons toutes les valeurs d'un marché dont le prix de clôture est strictement inférieur à la bande de Bollinger inférieure. Les bandes de Bollinger sont calculées sur 10 périodes et appliquées au cours de clôture.

Il est possible d'étendre la recherche à plusieurs conditions, qui peuvent être satisfaites en même temps ou alternativement. Pour ce faire, la syntaxe de ProScreener sera :

```
SCREENER[Condition1 AND Condition2]
ou bien
SCREENER[Condition1 OR Condition2]
```

Exemple :

Recherchons les valeurs dont le prix est supérieur au canal de Bollinger et qui présentent une tendance à la hausse.

```
// Clôture au dessus de la Boll sup
Condition1 = (Close > BollingerUp[20](Close))
// Clôture > à l'Ouverture
Condition2 = Close > Open
// MM7 > MM23
Condition3 = (Average[7](Close) > Average[23](Close))
SCREENER[Condition1 AND Condition2 AND Condition3]
```

Il est également possible de définir une constante ou un indicateur (prédéfini ou personnalisé) qui servira à la fois comme critère de filtre et de tri des résultats. En effet :

Si plus de 50 ou 100 résultats correspondent à la recherche, l'indicateur filtre les valeurs à afficher. Dans ce cas, la troisième section de la fenêtre de programmation vous permet d'en définir l'ordre de parution (voir [description page 5](#)).

Si moins de 50 ou 100 résultats correspondent à la recherche, l'indicateur trie les résultats sur la base de ce nouvel indicateur.

La syntaxe pour utiliser une constante est :

```
SCREENER[c1] (Criteria)
```

Pour un indicateur prédéfini, il est préférable de stocker d'abord l'indicateur dans une variable (ici on l'appellera "Criteria") :

```
Criteria = RSI[14] (Close)
SCREENER[c1] (Criteria)
```

Si on fait appel à un indicateur personnalisé, il faudra utiliser l'instruction "CALL", définie dans le [manuel ProBuilder](#).

```
MyRSI = CALL "RSI" [14]
SCREENER[c1] (MyRSI)
```

#### Exemple :

Repérons toutes les valeurs d'un marché dont le volume est supérieur à 50000. Créons un indicateur qui nous permet d'évaluer si le RSI est en surachat ou en survente. Nous afficherons les résultats du ProScreener triés sur la base de ce dernier indicateur.

```
c1 = Volume > 50000
IF RSI[20] (Close) > 70 THEN
    Criteria = 1
ELSIF RSI[20] (Close) < 30 THEN
    Criteria = -1
ENDIF
SCREENER[c1] (Criteria AS "sur-transaction")
```

Nom	%Var	Volume	sur-transaction
HERMES INTL	+2,59 %	24 537	1
AIR LIQUIDE	+1,23 %	203 970	1
CARREFOUR	+1,24 %	941 057	1
TOTALENERGIES	+0,06 %	2 684k	1
LOREAL	+2,86 %	216 801	1
SANOFI	+0,20 %	552 490	1
PERNOD RICARD	+1,75 %	148 551	1
THALES	-1,31 %	115 620	1
SCHNEIDER ELECTRIC	+2,37 %	246 700	1
CAPGEMINI	+0,99 %	95 953	1
ORANGE	-0,98 %	1 960k	1
AIRBUS	+0,52 %	435 657	1
EUROFINS SCIENT.	+0,07 %	127 308	1
SAFRAN	+0,83 %	213 556	1
UNIBAIL-RODAMCO-WE	-2,37 %	449 404	1
ARCELORMITTAL SA	+2,41 %	2 473k	1
EUROAPI	+1,78 %	330 508	0
CREDIT AGRICOLE	+0,28 %	1 820k	-1

Nb résultats : 41 (max 100 affichés)      Quantité d'historique : 1024

Il est possible de ne pas utiliser les parenthèses, en introduisant la commande "SORT BY" :

```
SCREENER[c1] SORT BY Criteria
```

Si nous souhaitons attribuer un nom personnalisé à cette colonne, nous devons utiliser la fonction "AS", comme illustré ci-dessous :

```
SCREENER[c1] (Criteria AS "typical price")
```

### Utiliser plusieurs critères de tri

Le screener permet également de définir plusieurs critères de tri dans le code afin d'afficher des valeurs complémentaires dans la fenêtre de résultat et pour être capable ensuite de modifier le critère de tri d'un code au lancement sans avoir besoin de modifier le code.

Les critères de tri au sein d'une instructions screener seront séparé par des virgules

**Exemple :**

```
SCREENER[c1] (Criteria AS "sur-transaction",Criteria2 AS  
"achat/vente",number AS "Format Number" DATEFORMAT)
```

Cette fonctionnalité ne sera disponible que pour les codes ne contenant qu'une seule instruction **SCREENER**.

### Formater les données de sortie

Le screener donne aussi la possibilité de préciser quel sera le type de données affichées grâce aux instructions de format :

- **PERCENTFORMAT** : Affiche les valeurs de la colonne sous forme de pourcentage
- **STRINGFORMAT** : Affiche les valeurs de la colonne sans formatage
- **NUMBERFORMAT** : Affiche les valeurs de la colonne au format numérique
- **DATEFORMAT** : Affiche les valeurs de la colonne sous la forme d'une date

Prenons le nombre suivant :

```
number = 20221202
```

Voici les résultats affichés pour les différents types de format proposés :

Format	Résultat
SCREENER(number AS "Format Number" PERCENTFORMAT)	2 022 120 200,00%
SCREENER(number AS "Format Number" STRINGFORMAT)	20221202
SCREENER(number AS "Format Number" NUMBERFORMAT)	20,2M
SCREENER(number AS "Format Number" DATEFORMAT)	2 déc 2022

## Anticipation du volume

La commande "**EstimatedVolume**" permet d'avoir une estimation linéaire du volume de la barre courante. Plus précisément, on calcule le volume estimé à partir de la formule suivante :

$$\text{Volume\_estimé} = \text{Volume} * \text{Coefficient\_Multiplicateur}$$

où

$$\text{Coefficient\_Multiplicateur} = \text{durée de la vue} / \text{temps écoulé depuis que le chandelier a commencé}$$

Cette commande se révèle donc particulièrement intéressante pour comparer le volume estimé au volume réel.

Regardons un exemple :

En vue 10 minutes, si l'on suppose que le dernier chandelier se dessine depuis 1 minute, le volume estimé sera égal à 10 fois le volume actuel.

Cherchons à calculer le rapport du volume anticipé du jour sur le volume de la veille (pour avoir des résultats représentatifs, utiliser une période intraday) :

```
// Evaluate le volume du jour
Vo10 = EstimatedVolume
// Trouve le volume de la veille
Vo11 = Volume[1]
// Critère de tri : rapport du volume anticipé du jour sur le volume de la veille
SCREENER SORT BY Vo10 / Vo11 AS "Volume"
```

## Recherche multi-périodes

Il vous est possible de définir une recherche qui vérifie les critères dans plusieurs unités de temps. Ceci permet de vérifier qu'une condition soit satisfaite par exemple dans le court et le moyen terme. La commande adaptée est "**TIMEFRAME**", et sa syntaxe est la suivante :

**TIMEFRAME** (CODE de l'unité de temps de la période)

Les différentes unités de temps disponibles se traduisent en code par :

SIGNIFICATION	CODE
1 minute	1 minute
2 minutes	2 minutes
3 minutes	3 minutes
5 minutes	5 minutes
10 minutes	10 minutes
15 minutes	15 minutes
30 minutes	30 minutes
1 heure	1 hour
2 heures	2 hours
3 heures	3 hours
4 heures	4 hours
Journalier	Daily
Hebdomadaire	Weekly
Mensuel	Monthly
Trimestriel	Quarterly
Annuel	Yearly

Les instructions qui suivent la ligne "TIMEFRAME" seront calculées seulement dans la période indiquée. Il est donc possible de placer plusieurs lignes "TIMEFRAME" dans un même code, afin d'effectuer une recherche multi-périodes.

Prenons un exemple :

On souhaite repérer toutes les valeurs d'Euronext Paris qui vérifient les critères ci-dessous :

- En vue hebdomadaire, l'indicateur Williams %R à 14 périodes a une valeur comprise entre 0 et -20
- En vue 30 minutes, la moyenne mobile exponentielle à 20 périodes vient de croiser à la hausse celle à 12 périodes.

Nous pourrions donc détecter les valeurs en zone de surachat en hebdomadaire, mais qui présentent à court terme une tendance baissière.

```
TIMEFRAME(weekly)
Condition1 = Williams[14](Close) < 0 AND Williams[14](Close) > -20
TIMEFRAME(30 minutes)
Condition2 = ExponentialAverage[20](Close) CROSSES OVER ExponentialAverage[12](Close)
SCREENER[Condition1 AND Condition2]
```

Pour revenir aux données de la période sélectionnée dans l'interface ProScreener, nous pouvons écrire :

```
TIMEFRAME (default)
```

## Recherche multi-valeurs

---

Nous avons pu constater qu'avec ProScreener il est nécessaire de spécifier un marché (ou une liste) sur lequel effectuer la recherche.

L'instruction "EQUITYFRAME", permet de mettre en relation le critère avec une valeur spécifique .

Elle s'utilise comme suit :

```
EQUITYFRAME ("nom du marché", "ticker")
```

La commande permet donc de comparer des résultats à une valeur particulière, mais aussi de construire un nouvel indicateur utilisé comme condition de tri.

Par exemple, si vous souhaitez afficher 50 valeurs du marché Euronext Paris, avec comme critère le prix de clôture de la barre courante supérieur à celle de la valeur AXA (ticker "CS") , vous écrirez :

```
MyClose = Close
EQUITYFRAME("Euronext Paris", "CS")
AXAClose = Close
SCREENER MyClose>AXAClose SORT BY MyClose as "Close"
```

Vous pouvez retrouver les noms des marchés et les tickers dans la fenêtre « Recherche d'instrument ». Lors de l'exécution de votre screener, si une erreur de nom s'est glissée dans votre commande EQUITYFRAME, la plateforme ouvrira automatiquement la recherche d'instrument pour vous permettre de sélectionner celui-ci.

Comme pour la commande `TIMEFRAME`, tout code qui suivra `EQUITYFRAME` sera appliqué à la valeur spécifiée par ce dernier. Pour revenir aux données du marché sélectionné dans l'interface ProScreener, nous pouvons écrire :

`EQUITYFRAME (default)`

Les deux utilisations présentées ci-dessus sont complémentaires et l'exemple suivant l'illustre bien :

Si vous souhaitez afficher sur 50 ou 100 valeurs prises au hasard du marché Euronext Paris, l'indicateur faisant la différence du prix de clôture de la barre courante de la valeur AXA (ticker "CS") par le prix de clôture respectif des valeurs sélectionnées

```
// on commence par récupérer les informations à extraire de la valeur AXA d'Euronext
Paris
EQUITYFRAME("Actions France", "CS")
MyClose = Close
// on revient sur les valeurs du marché sélectionné dans le menu déroulant "Sélection de
la liste"
EQUITYFRAME(default)
// on récupère les informations que l'on souhaite
CloseVal = MyClose - Close
// on lance l'affichage
SCREENER SORT BY CloseVal AS "MyIndicator"
```

Exemple : Ce screener permet de représenter visuellement la corrélation entre un titre et son marché. On calcule le ratio des deux valeurs sélectionnées. On calcule ensuite la différence de ce ratio par rapport à la veille.

```
TIMEFRAME(daily)
CloseVal = Close
EQUITYFRAME("Actions NASDAQ", "AMZN")
CloseInd = Close
EQUITYFRAME(default)
Ratio = (CloseVal / CloseInd) * 100
ForceRelative = (Ratio - Ratio[1]) * 100
SCREENER SORT BY ForceRelative AS "Force Relative"
```

## Chapitre III : Applications pratiques

### Exemples simples

---

#### RSI 1heure : survente

Le RSI est un indicateur de surachat/survente qui peut être précurseur d'un retournement de tendance. Nous allons ici détecter les valeurs qui sont en survente.

La survente est définie par : **RSI < 30**. Elle est d'autant plus marquée que le RSI est proche de zéro. Nous allons donc construire un ProScreener qui renvoie toutes les valeurs dont le RSI est inférieur à 30.

```
// Calcule le RSI à 14 barres
MyRSI = RSI[14]
// Filtre : RSI < 30
Filtre = MyRSI < 30
SCREENER[Filtre] SORT BY MyRSI AS "RSI"
```

#### RSI 1heure : surachat

Recherchons les valeurs en surachat (défini par : **RSI > 70**). Il est d'autant plus marqué que le RSI est proche de 100. Nous allons donc construire un screener qui renvoie toutes les valeurs dont le RSI est supérieur à 70.

```
// Calcule le RSI à 14 barres
MyRSI = RSI[14]
// Filtre : RSI > 70
Filtre = MyRSI > 70
SCREENER[Filtre] SORT BY MyRSI AS "RSI"
```

#### Croisement de moyennes mobiles haussier

Il s'agit de mettre en évidence une tendance où une moyenne mobile courte croise une moyenne mobile longue.

Nous allons construire un ProScreener qui renvoie toutes les valeurs dont la moyenne mobile à 20 jours croise à la hausse la moyenne mobile à 50 jours.

On calcule par ailleurs le momentum de la différence entre les deux moyennes mobiles afin d'évaluer la puissance du croisement. Si ce nombre est proche de zéro, le croisement est lent et les deux moyennes sont presque parallèles donnant un signal peu significatif. Au contraire, plus ce nombre est élevé et plus le croisement est "dynamique". Un exemple typique est celui d'une moyenne mobile plate croisant une moyenne mobile très nettement orientée à la hausse.

```
// Calcul de la moyenne mobile à 20 jours
Sma20 = Average[20]
REM Calcul de la moyenne mobile à 50 jours
Sma50 = Average[50]
// Evalue la vitesse relative de la moyenne mobile courte par rapport à la longue
Vitesse = Momentum(Sma50 - Sma20) * 100 / Close
// Sélectionne la valeur au moment du croisement
Filtre = Sma20 CROSSES OVER Sma50
SCREENER[Filtre] SORT BY Vitesse AS "Dynamique"
```

## Croisement de moyennes mobiles baissier

Nous allons construire un ProScreener qui renvoie toutes les valeurs dont la moyenne mobile à 20 jours croise à la baisse la moyenne mobile à 50 jours.

On définit le momentum de la même façon que sur l'exemple précédent.

```
// Calcul de la moyenne mobile à 20 jours
Sma20 = Average[20]
// Calcul de la moyenne mobile à 50 jours
Sma50 = Average[50]
// Evalue la vitesse relative de la moyenne mobile courte par rapport à la longue
Vitesse = Momentum(Sma20 - Sma50) * 100 / Close
// Sélectionne la valeur au moment du croisement
Filtre = Sma20 CROSSES UNDER Sma50
SCREENER[Filtre] SORT BY Vitesse AS "Dynamique"
```

## Exemples plus élaborés

---

### RSI et retournement haussier

Nous vous présentons ici un système permettant de détecter en temps réel les valeurs les plus susceptibles de se retourner et d'amorcer un mouvement de grande envergure.

Traditionnellement les analystes s'intéressaient aux indicateurs de surchauffe ou aux figures de retournement en les isolant de leur contexte (simplement parce qu'ils ne disposaient pas des outils techniques leur permettant de faire mieux).

Aujourd'hui avec le ProScreener il n'y a plus de limitation technique, nous avons donc la possibilité de construire un véritable système de détection basé sur le RSI. Le point de départ est simple : on cherche un RSI en survente qui se retourne à la hausse.

On met donc en place un filtre contenant ces deux conditions ce qui s'écrit en une seule ligne de code :

```
Filtre = RSI < 30 AND Momentum[1](RSI) > 0
```

Maintenant que nous avons les valeurs susceptibles de nous intéresser, nous allons les classer pour mettre en évidence les placements les plus intéressants. Pour cela, on considère ceux dont la chute a été la plus sévère, comparativement à la volatilité habituelle du titre.

```
// Filtre les valeurs dont le RSI est en survente et en retournement
// Critère de filtrage : RSI < 30 et croissant
Filtre = RSI < 30 AND Momentum[1](RSI) > 0
// Détermine la "force" de la tendance baissière
// Trouve le point le plus haut sur les 20 barres antérieures
PointHaut = highest[20](High)
// Détermine le déclin depuis ce point
Declin = PointHaut - Close
// Détermine la volatilité habituelle du titre (médiane du True Range sur 3 barres)
Norme = summation[3](TR) - highest[3](TR) - lowest[3](TR)
Vitesse = Declin / Norme
// Affichage des résultats
SCREENER[Filtre] SORT BY Vitesse AS "Puissance baissière"
```

## RSI et retournement baissier

Le point de départ de ce ProScreener est la recherche d'un RSI en surachat qui se retourne à la baisse.

Comme précédemment, mettons en place un filtre. Celui-ci s'écrit en code ProBuilder :

```
Filtre = RSI > 70 AND Momentum[1](RSI) < 0
```

De façon analogue au ProScreener précédent, on récupérera les valeurs dont la hausse a été la plus marquante, comparativement à la volatilité habituelle du titre.

```
// Filtre les valeurs dont le RSI est en surachat et en retournement
// Critère de filtrage : RSI > 70 et décroissant
Filtre = RSI > 70 AND Momentum[1](RSI) < 0
// Détermine la "force" de la tendance haussière
// Trouve le point le plus bas sur les 20 barres antérieures
PointBas = lowest[20](Low)
// Détermine la progression des cours depuis ce point
Hausse = Close - PointBas
// Détermine la volatilité habituelle du titre (médiane du True Range sur 3 barres)
Norme = summation[3](TR) - highest[3](TR) - lowest[3](TR)
Vitesse = Hausse / Norme
// Palmarès de screening : les hausses relatives les plus marquantes
SCREENER[Filtre] SORT BY Vitesse AS "Puissance haussiere"
```

## Avalement haussier

L'avalement haussier est une des figures en chandeliers japonais les plus connues mais elle nécessite évidemment la prise en compte du contexte (existence d'une tendance à retourner).

Commençons par définir ce qu'est un avalement haussier :

- Chandelier précédent de couleur noire
- Chandelier courant ouvrant en-dessous du corps du chandelier précédent
- Chandelier courant clôturant au-dessus du corps du chandelier précédent

Ces trois conditions s'écrivent simplement :

```
Filtre = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]
```

Pour ce qui est de la détection de la tendance baissière, nous reprenons le code précédent en resserrant un petit peu l'horizon de temps. En effet, avec le RSI **14**, il était raisonnable de s'intéresser à la chute des cours intervenue sur les 20 barres antérieures. Pour ce qui est des chandeliers japonais, l'expérience montre que 8 barres sont suffisantes (ce sont des structures plus réactives et théoriquement limitées à 7 chandeliers).

```
// Détermine la "force" de la tendance baissière
// Trouve le point le plus haut sur les 8 barres antérieures
PointHaut = highest[8](High)
// Détermine le déclin depuis ce point
Declin = PointHaut - Close
// Détermine la volatilité habituelle du titre (médiane du True Range sur 3 barres)
Norme = summation[3](TR) - highest[3](TR) - lowest[3](TR)
// Critère de filtrage : avalement haussier
Filtre = Close[1] < Open[1] AND Open < Close[1] AND Close > Open[1]
Vitesse = Declin / Norme
// Palmarès de screening : les déclins relatifs les plus sévères
SCREENER[Filtre] SORT BY Vitesse AS "Puissance du déclin"
```

## Avalement baissier

Commençons par définir ce qu'est un avalement baissier :

- Chandelier précédent de couleur blanche
- Chandelier courant ouvrant au-dessus du corps du chandelier précédent
- Chandelier courant clôturant en-dessous du corps du chandelier précédent

Ces trois conditions s'écrivent simplement :

```
Filtre = Close[1] > Open[1] AND Open > Close[1] AND Close < Open[1]
```

Pour ce qui est de la détection de la tendance haussière, nous reprenons le code du RSI en retournement baissier.

```
// détermine la "force" de la tendance haussière
// Trouve le point le plus bas sur les 8 barres antérieures
PointBas = lowest[8](Low)
// Détermine la progression des cours depuis ce point
Hausse = Close - PointBas
// Détermine la volatilité habituelle du titre (médiane du True Range sur 3 barres)
Norme = summation[3](TR) - highest[3](TR) - lowest[3](TR)
// Critère de filtrage : avalement baissier
Filtre = Close[1] > Open[1] AND Open > Close[1] AND Close < Open[1]
Vitesse = Hausse / Norme
// Palmarès de screening : les hausses relatives les plus marquantes
SCREENER[Filtre] SORT BY Vitesse AS "Puissance haussière"
```

## Triple écran haussier

Ce ProScreener est composé de trois conditions sur plusieurs unités de temps :

- **Condition 1** : MACD en vue hebdomadaire < 0
- **Condition 2** : MACD en vue hebdomadaire croissant
- **Condition 3** : Stochastique en vue journalière < 30

On cherche les valeurs plus proches du niveau de l'achat STOP préconisé par le système (dans ce cas, le cours le plus haut de la veille).

Le ProScreener affiche les valeurs qui se trouvent encore sous ce niveau, et celles qui viennent de le franchir et qui restent à proximité (pas plus de +5%).

Les résultats affichés respectent les conditions dans les unités de temps spécifiées (vue hebdomadaire et journalière).

```
// Condition 1 : MACD weekly < 0 et croissant
TIMEFRAME(weekly)
MyMACD = MACD[12, 26, 9](Close)
c1 = MyMACD < 0 AND MyMACD > MyMACD[1]
// Condition 2 : Stochastic daily < 30
TIMEFRAME(daily)
MySTO = Stochastic[14, 3](Close)
c2 = MySTO < 30
// Niveau de l'achat stop
MyStop = High[1]
// Critère : Position des cours par rapport au STOP
Criteria = (Close / MyStop - 1) * 100
// Condition 3 : les cours sous le STOP ou à moins de +5%
c3 = Criteria < 5
SCREENER[c1 AND c2 AND c3] SORT BY Criteria
```

## Triple écran baissier

Ce ProScreener est composé de trois conditions sur plusieurs unités de temps :

- **Condition 1** : MACD en vue hebdomadaire > 0
- **Condition 2** : MACD en vue hebdomadaire décroissant
- **Condition 3** : Stochastic en vue journalière > 70

On cherche les valeurs qui sont les plus proches du niveau de la vente STOP préconisée par le système. Ce niveau est le cours le plus bas de la veille.

Le ProScreener affiche les valeurs qui sont encore au-dessus de ce niveau, et celles qui viennent de le franchir et qui restent à proximité (pas plus de -5%), dans le deux unités de temps spécifiées.

```
// Condition 1 : MACD weekly > 0 et décroissant
TIMEFRAME(weekly)
MyMACD = MACD[12,26,9](Close)
c1 = MyMACD > 0 AND MyMACD < MyMACD[1]
// Condition 2 : Stochastic daily > 70
TIMEFRAME(daily)
MySTO = Stochastic[14,3](Close)
c2 = MySTO > 70
// Niveau de la vente stop
MyStop = Low[1]
// Critère : Position des cours par rapport au STOP
Criteria = (Close / MyStop - 1) * 100
// Condition 3 : les cours au-dessus du STOP ou à moins de +5%
c3 = Criteria > -5
SCREENER[c1 AND c2 AND c3] SORT BY Criteria
```

Vous pouvez consulter notre communauté ProRealTime sur le [forum ProRealCode](#) afin d'y retrouver une [documentation en ligne](#) ainsi que de nombreux exemples.

## Glossaire

## A

CODE	IMPLÉMENTATION	FONCTION
ABS	ABS(a)	Fonction Mathématique "Valeur Absolue"
AccumDistr	AccumDistr(price)	Désigne l'Accumulation Distribution classique
ADX	ADX[N]	Indicateur Average Directional Index
ADXR	ADXR[N]	Indicateur Average Directional Index Rate
AND	a AND b	Opérateur logique ET
ArrayMax	ArrayMax(\$MyArray)	Retourne la valeur la plus grande
ArrayMin	ArrayMin(\$MyArray)	Retourne la valeur la plus petite
ArraySort	ArraySort(\$MyArray, ASCEND)	Tri le tableau par ordre croissant (ASCEND) ou décroissant (DESCEND)
AroonDown	AroonDown[P]	Désigne l'Aroon Down
AroonUp	AroonUp[P]	Désigne l'Aroon Up
ATAN	ATAN(a)	Fonction mathématique "Arc tangente"
AS	RETURN x AS "ResultName"	Instruction servant à nommer une courbe
Average	Average[N](price)	Moyenne Mobile Arithmétique
AverageTrueRange	AverageTrueRange[N](price)	Désigne la moyenne mobile par lissage de Wilder du True Range

## B

CODE	IMPLÉMENTATION	FONCTION
BarIndex	BarIndex	Nombre de barres depuis la première barre de données chargée (dans un graphique dans le cas d'un indicateur ProBuilder ou pour un système de trading dans le cas d'un ProBacktest ou ProInvest)
BollingerBandWidth	BollingerBandWidth[N](price)	Bande passante de Bollinger
BollingerDown	BollingerDown[N](price)	Support de la bande de Bollinger
BollingerUp	BollingerUp[N](price)	Résistance de la bande de Bollinger
BREAK	(FOR...DO...BREAK...NEXT) ou (WHILE...DO...BREAK...WEND)	Instruction de sortie forcée de boucle FOR ou WHILE

## C

CODE	IMPLÉMENTATION	FONCTION
CALL	myResult = CALL myFunction	Appel de fonction utilisateur
CCI	CCI[N](price) ou CCI[N]	Donne le Commodity Channel Index
ChaikinOsc	ChaikinOsc[Ch1, Ch2](price)	Désigne l'oscillateur de Chaikin
Chandle	Chandle[N](price)	Désigne le Chande Momentum Oscillator
ChandeKrollStopUp	ChandeKrollStopUp[Pp, Qq, X]	Stop de protection selon Chande et Kroll en position acheteuse
ChandeKrollStopDown	ChandeKrollStopDown[Pp, Qq, X]	Stop de protection selon Chande et Kroll en position vendeuse
Close	Close[N]	Désigne le prix de clôture de la barre courante ou de celle n jours auparavant
COLOURED	RETURN x COLOURED(R,G,B)	Colorie une courbe d'une certaine couleur selon la convention RGB
COS	COS(a)	Fonction cosinus
CROSSES OVER	a CROSSES OVER b	Opérateur booléen vérifiant qu'une courbe passe au-dessus d'une autre
CROSSES UNDER	a CROSSES UNDER b	Opérateur booléen vérifiant qu'une courbe passe en dessous d'une autre
cumsum	cumsum(price)	Sommation d'un prix depuis le début de l'historique affiché
CurrentDayOfWeek	CurrentDayOfWeek	Désigne le jour actuel
CurrentHour	CurrentHour	Désigne l'heure actuelle
CurrentMinute	CurrentMinute	Désigne la minute actuelle
CurrentMonth	CurrentMonth	Désigne le mois actuel
CurrentSecond	CurrentSecond	Désigne la seconde actuelle
CurrentTime	CurrentTime	Désigne HeureMinute actuelle
CurrentYear	CurrentYear	Désigne l'année actuelle
CustomClose	CustomClose[N]	Constante paramétrable dans la fenêtre de propriétés
Cycle	Cycle(price)	Indicateur Cycle

## D

CODE	IMPLÉMENTATION	FONCTION
Date	Date[N]	Désigne la date de clôture de la barre courante
DATEFORMAT	SCREENER(date DATEFORMAT)	Affiche les valeurs de la colonne sous la forme d'une date
Day	Day[N]	Jour de clôture de la barre courante
Days	Days[N]	Compteur de jours depuis 1900
Daily	TIMEFRAME(Daily)	Définit la période « journalier » pour la suite des calculs dans le code du screener.
DayOfWeek	DayOfWeek[N]	Désigne le jour de la semaine durant lequel la barre courante a clos
DClose	DClose(N)	Prix de clôture de la n-ième journée antérieure à celle de la barre courante
DEMA	DEMA[N](price)	Double Moyenne Mobile Exponentielle
DHigh	DHigh(N)	Prix le plus haut de la n-ième journée antérieure à celle de la barre courante
DI	DI[N](price)	Désigne le Demand Index
DIminus	DIminus[N](price)	Désigne le DI-
DIplus	DIplus[N](price)	Désigne le DI+
DLow	DLow(N)	Prix le plus bas de la n-ième journée antérieure à celle de la barre courante
DO	Voir FOR et WHILE	Instruction facultative des FOR et WHILE pour l'action de bouclage
DOpen	DOpen(N)	Prix d'ouverture de la n-ième journée antérieure à celle de la barre courante
DOWNTO	Voir FOR	Instruction sur boucle FOR pour une lecture décroissante
DPO	DPO[N](price)	Désigne le Detrended Price Oscillator

## E

CODE	IMPLÉMENTATION	FONCTION
EaseOfMovement	EaseOfMovement[I]	Désigne l'indicateur Ease of Movement
ELSE	Voir IF/THEN/ELSE/ENDIF	Instruction d'appel de la seconde condition à défaut de la première issue du IF
ELSEIF	Voir IF/THEN/ELSE/ENDIF	Contraction de ELSE IF
EMV	EMV[N]	Désigne l'indicateur Ease of Movement Value
EQUITYFRAME	EQUITYFRAME("marché", "ticker")	Recherche une condition en relation avec une valeur du meme marché
ENDIF	Voir IF/THEN/ELSE/ENDIF	Instruction de clôture des instructions conditionnelles
EndPointAverage	EndPointAverage[N](price)	Moyenne Mobile à dernier point
EstimatedVolume	EstimatedVolume	Indique un volume anticipé
EXP	EXP(a)	Fonction Mathématique "Exponentielle"
ExponentialAverage	ExponentialAverage[N](price)	Moyenne Mobile Exponentielle

## F - G

CODE	IMPLÉMENTATION	FONCTION
FOR/TO/NEXT	FOR i =a TO b DO a NEXT	Prend les valeurs désignées du début à la fin ou vice versa
ForceIndex	ForceIndex(price)	Indicateur Force Index déterminant qui contrôle le marché

## H

CODE	IMPLÉMENTATION	FONCTION
High	High[N]	Désigne le plus haut cours atteint durant la période N
Highest	Highest[N](price)	Désigne le plus haut cours sur un horizon donné
HistoricVolatility	HistoricVolatility[N](price)	Désigne la volatilité historique ou statistique
Hour	Hour[N]	Désigne l'heure de clôture de chaque barre
Hours	TIMEFRAME(X Hours)	Définit la période « X heure » pour la suite des calculs dans le code du screener. (entre 1 et 4, voir <a href="#">Recherche Multi-période</a> )

## I - J - K

CODE	IMPLÉMENTATION	FONCTION
IF/THEN/ENDIF	IF a THEN b ENDIF	Ensemble d'instructions conditionnelles sans deuxième condition
IF/THEN/ELSE/ENDIF	IF a THEN b ELSE c ENDIF	Ensemble d'instructions conditionnelles
IntradayBarIndex	IntradayBarIndex[N]	Compte le nombre de chandeliers sur le graphique intraday
IsSet	IsSet(\$MyArray[index])	Retourne 1 si l'indice (index) du tableau a été défini, 0 si il n'a pas été défini.

## L

CODE	IMPLÉMENTATION	FONCTION
LastSet	LastSet(\$MyArray)	Retourne l'index défini le plus élevé du tableau
LinearRegression	LinearRegression[N](price)	Droite de régression linéaire
LinearRegressionSlope	LinearRegressionSlope[N](price)	Pente de la droite de régression linéaire
LOG	LOG(a)	Fonction mathématique "logarithme népérien"
Low	Low[N]	Désigne le plus bas atteint durant la période
Lowest	Lowest[N](price)	Désigne le plus bas d'une période sur un horizon donné

## M

CODE	IMPLÉMENTATION	FONCTION
MACD	MACD[S,L,Si](price)	Moving Average Convergence Divergence (MACD)
MACDline	MACDLine[S,L](price)	Désigne la ligne du MACD
MassIndex	MassIndex[N]	Indicateur Mass Index appliqué sur N barres
MAX	MAX(a,b)	Fonction mathématique "Maximum"
MedianPrice	MedianPrice	Moyenne du prix le plus haut et du plus bas
MIN	MIN(a,b)	Fonction Mathématique "Minimum"
Minute	Minute	Désigne la minute du moment de la clôture de chaque barre de l'historique
MOD	a MOD b	Fonction Mathématique "Reste de la division euclidienne"
Momentum	Momentum[I]	Momentum (prix de clôture – prix de clôture de la n-ième barre précédente)
MoneyFlow	MoneyFlow[N](price)	Donne le MoneyFlow entre -1 et 1
MoneyFlowIndex	MoneyFlowIndex[N]	Désigne le MoneyFlowIndex

CODE	IMPLÉMENTATION	FONCTION
Month	Month[N]	Désigne le mois de la clôture de chaque barre de l'historique
Monthly	TIMEFRAME(Monthly)	Définit la période « mensuel » pour la suite des calculs dans le code du screener.

## N

CODE	IMPLÉMENTATION	FONCTION
NEXT	Voir FOR/TO/NEXT	Instruction à placer à la fin de la boucle "FOR"
NOT	NOT a	Opérateur logique NON
NUMBERFORMAT	SCREENER(number NUMBERFORMAT)	Affiche les valeurs de la colonne sous la forme d'un nombre

## O

CODE	IMPLÉMENTATION	FONCTION
OBV	OBV(price)	Désigne l' "On-Balance-Volume"
ONCE	ONCE VariableName = VariableValue	Instruction qui en précède une autre qu'on ne veut réaliser qu'une seule fois
Open	Open[N]	Désigne le prix d'ouverture de la barre courante ou celle de n jours auparavant
OR	a OR b	Opérateur logique OU

## P - Q

CODE	IMPLÉMENTATION	FONCTION
PERCENTFORMAT	SCREENER(var PERCENTFORMAT)	Affiche les valeurs de la colonne sous la forme d'un pourcentage
PriceOscillator	PriceOscillator[S,L](price)	Indicateur Pourcentage Price oscillator
PositiveVolumeIndex	PriceVolumeIndex(price)	Désigne l'indicateur Positive Volume Index
PVT	PVT(price)	Désigne l'indicateur "Price Volume Trend"
Quarterly	TIMEFRAME(Quarterly)	Définit la période « trimestriel » pour la suite des calculs dans le code du screener.

## R

CODE	IMPLÉMENTATION	FONCTION
R2	R2[N](price)	Coefficient R Carré (taux d'erreur des prix à la regression linéaire)
Range	Range[N]	Différence entre le prix le plus haut et le plus bas de la barre courante

CODE	IMPLÉMENTATION	FONCTION
Repulse	Repulse[N](price)	Mesure la poussée haussière et baissière de chaque bougie
RETURN	RETURN Result	Instruction qui renvoie le résultat
ROC	ROC[N](price)	Désigne le "Price Rate of Change"
RSI	RSI[N](price)	Désigne l'oscillateur "Relative Strength Index"
ROUND	ROUND(a)	Fonction mathématique "Arrondi à l'unité"

## S

CODE	IMPLÉMENTATION	FONCTION
SAR	SAR[At,St,Lim]	Désigne le Parabolique SAR
SARatdmf	SARatdmf[At,St,Lim](price)	Pareil (parabolique SAR) Désigne le Parabolique SAR ATDMF
SCREENER	SCREENER[c](prix)	Affiche les résultats
SIN	SIN(a)	Fonction Mathématique "Sinus"
SGN	SGN(a)	Fonction Mathématique "Signe de"
SMI	SMI[N,SS,DS](price)	Désigne le Stochastic Momentum Index
SmoothedStochastic	SmoothedStochastic[N,K] (price)	Désigne une Stochastique lissée
SORT BY	Screeener(c1) SORT BY price	Permet l'affichage ou le tri des résultats
SQUARE	SQUARE(a)	Fonction mathématique "Mise au carré"
SQRT	SQRT(a)	Fonction Mathématique "Mise à la racine carrée"
STD	STD[N](price)	Fonction Statistique "écart-type"
STE	STE[N](price)	Fonction Statistique "écart-erreur"
STRINGFORMAT	SCREENER(number STRINGFORMAT)	Affiche les valeurs de la colonne sans formatage
Stochastic	Stochastic[N,K](price)	Ligne %K de la Stochastique
Summation	Summation[N](price)	Somme d'un certain prix des N derniers chandeliers
Supertrend	Supertrend[STF,N]	Désigne le Super Trend

## T

CODE	IMPLÉMENTATION	FONCTION
TAN	TAN(a)	Fonction mathématique "Tangente"

TEMA	TEMA[N](price)	Moyenne Mobile Exponentielle Triple
THEN	Voir IF/THEN/ELSE/ENDIF	Instruction suivant la première condition de l'instruction "IF"
Time	Time[N]	Donne l'HeureMinuteSeconde et permet de faire appel à l'heure
TimeSeriesAverage	TimeSeriesAverage[N](price)	Moyenne mobile des séries temporelles
TIMEFRAME	TIMEFRAME(P)	Permet de travailler sur une période différente de celle par défaut
TO	Voir FOR/TO/NEXT	Instruction "jusqu'à" dans la boucle "Pour"
Today	Today	Date de la journée actuelle
TotalPrice	TotalPrice[N]	(Clôture + Ouverture + Plus Haut + Plus Bas) / 4
TR	TR(price)	Désigne le True Range
TriangularAverage	TriangularAverage[N](price)	Moyenne Mobile Triangulaire
TRIX	TRIX[N](price)	Triple Moyenne Mobile Exponentielle
TypicalPrice	TypicalPrice[N]	Prix Typique (moyenne de plus haut, plus bas et clôture)

## U

CODE	IMPLÉMENTATION	FONCTION
Undefined	a = Undefined	Pour laisser une variable indéfinie (Null)
UnSet	UnSet(\$MyArray)	Réinitialise les données du tableau

## V

CODE	IMPLÉMENTATION	FONCTION
Variation	Variation(price)	Différence entre la clôture de la veille et la clôture courante en %
Volatility	Volatility[S, L]	Désigne la volatilité de Chaikin
Volume	Volume[N]	Désigne le volume
VolumeOscillator	VolumeOscillator[S,L]	Désigne l'oscillateur de volume
VolumeROC	VolumeROC[N]	Désigne le volume du Rate Of Change

## W

CODE	IMPLÉMENTATION	FONCTION
Weekly	TIMEFRAME(Weekly)	Définit la période « semaine » pour la suite des calculs dans le code du screener.
WeightedAverage	WeightedAverage[N](price)	Désigne la Moyenne Mobile Pondérée

CODE	IMPLÉMENTATION	FONCTION
<code>WeightedClose</code>	<code>WeightedClose[N]</code>	Moyenne pondérée entre le prix de clôture, le plus haut et la plus bas
<code>WEND</code>	Voir <code>WHILE/DO/WEND</code>	Instruction à placer à la fin de la boucle Tant Que
<code>WHILE/DO/WEND</code>	<code>WHILE (condition) DO (action) WEND</code>	Boucle "Tant Que"
<code>WilderAverage</code>	<code>WilderAverage[N](price)</code>	Donne la moyenne mobile de Wilder
<code>Williams</code>	<code>Williams[N](close)</code>	Calcule le %R de Williams
<code>WilliamsAccumDistr</code>	<code>WilliamsAccumDistr(price)</code>	Indicateur Accumulation/Distribution de Williams

## X

CODE	IMPLÉMENTATION	FONCTION
<code>XOR</code>	<code>a XOR b</code>	Opérateur logique OU exclusif

## Y

CODE	IMPLÉMENTATION	FONCTION
<code>Year</code>	<code>Year[N]</code>	Donne l'évolution des années
<code>Yearly</code>	<code>TIMEFRAME(Yearly)</code>	Définit la période « annuel » pour la suite des calculs dans le code du screener.
<code>Yesterday</code>	<code>Yesterday[N]</code>	Donne l'évolution du jour d'avant

## Z

CODE	IMPLÉMENTATION	FONCTION
<code>ZigZag</code>	<code>ZigZag[Zr](price)</code>	Zig-Zag de la théorie des vagues d'Eliott
<code>ZigZagPoint</code>	<code>ZigZagPoint[Zp](price)</code>	Zig-Zag de la théorie des vagues d'Eliott calculé à Zp points

## Autres

CODE	FONCTION	CODE	FONCTION
<code>+</code>	Opérateur d'addition	<code>&lt;&gt;</code>	Opérateur de différence
<code>-</code>	Opérateur de soustraction	<code>&lt;</code>	Opérateur d'infériorité strict
<code>*</code>	Opérateur de multiplication	<code>&gt;</code>	Opérateur de supériorité strict
<code>/</code>	Opérateur de division décimale	<code>&lt;=</code>	Opérateur d'infériorité
<code>=</code>	Opérateur d'égalité	<code>&gt;=</code>	Opérateur de supériorité

# ProRealTime SOFTWARE

